```
>
```

http://www.mapleprimes.com/questions/204334-How-To-Calculate-That-Integral

```
> restart; interface(version); with(IntegrationTools):
```

Classic Worksheet Interface, Maple 18.00, Windows, Feb 10 2014, Build ID 922027

```
> Int(abs(x-(-x^5+1)^(1/5)), x = 0 .. 1):
  plot(op(%));
  %%;
  ``=Split(%, solve(x-(-x^5+1)^(1/5)));
  evalf(%);
```



$$\int_0^1 \left| x-(-x^5+1)^{(1/5)} \right| dx$$

$$= \int_0^{\frac{2^{(4/5)}}{2}} \left| x-(-x^5+1)^{(1/5)} \right| dx + \int_{\frac{2^{(4/5)}}{2}}^1 \left| x-(-x^5+1)^{(1/5)} \right| dx$$

$$= 0.500000000000000$$

```
>
> b = (-b^k+1)^(1/k);

  lhs(%)^k = rhs(%)^k:
  combine(%) assuming 0<k, 0<b,b<1:
  my:=isolate(%, b^k);
  beta:= simplify(solve(%, b));
```

$$b = (-b^k+1)^{\left(\frac{1}{k}\right)}$$

$$my := b^k = \frac{1}{2}$$

$$\beta := 2^{\left(-\frac{1}{k}\right)}$$

```
> -Int((x-(-x^k+1)^(1/k)), x = 0 .. b);
  Expand(%):
  W1:=value(%) assuming 0<k;
```

$$-\int_0^b x-(-x^k+1)^{\left(\frac{1}{k}\right)} \, dx$$

$$W1 := -\frac{b^2}{2} + b \operatorname{hypergeom}\left(\left[\frac{1}{k}, -\frac{1}{k}\right], \left[\frac{k+1}{k}\right], b^k\right)$$

```
> Int((x-(-x^k+1)^(1/k)), x = b .. 1);
  Expand(%):
  W2:=value(%) assuming 0<k;
```

$$\int_b^1 x-(-x^k+1)^{\left(\frac{1}{k}\right)} \, dx$$

$$W2 := -\frac{b^2}{2} + \frac{1}{2} + b \operatorname{hypergeom}\left(\left[\frac{1}{k}, -\frac{1}{k}\right], \left[\frac{k+1}{k}\right], b^k\right) - \frac{\Gamma\left(\frac{k+1}{k}\right)^2}{\Gamma\left(\frac{k+1}{k} + \frac{1}{k}\right)}$$

```
>
> '1/2 = W1+W2';
  isolate(%, hypergeom):
  simplify(%);

  eval(%, my): eval(%, b=beta):
  E1:=simplify(%);
```

$$\frac{1}{2} = W1 + W2$$

$$\mathrm{hypergeom}\left(\left[\frac{1}{k}, -\frac{1}{k}\right], \left[\frac{k+1}{k}\right], b^k\right) = \frac{b}{2} + \frac{1}{2}\frac{\Gamma\left(\frac{k+1}{k}\right)^2}{b\,\Gamma\left(\frac{2+k}{k}\right)}$$

$$E1 := \mathrm{hypergeom}\left(\left[\frac{1}{k}, -\frac{1}{k}\right], \left[\frac{k+1}{k}\right], \frac{1}{2}\right) = 2^{\left(-\frac{k+1}{k}\right)} + \frac{2^{\left(-\frac{k-1}{k}\right)}\Gamma\left(\frac{k+1}{k}\right)^2}{\Gamma\left(\frac{2+k}{k}\right)}$$

```
> # check some k
  eval(E1, k=5); evalf(%);
  #[seq(%, k=1 .. 6)]; evalf[2*Digits](%): fnormal(%): map(is, %);
```

$$\mathrm{hypergeom}\left(\left[\frac{-1}{5}, \frac{1}{5}\right], \left[\frac{6}{5}\right], \frac{1}{2}\right) = \frac{2^{(4/5)}}{4} + \frac{1}{20}\frac{2^{(1/5)}\pi\csc\left(\frac{\pi}{5}\right)^2\Gamma\left(\frac{3}{5}\right)}{\Gamma\left(\frac{4}{5}\right)^2\csc\left(\frac{2\pi}{5}\right)}$$

$$0.980993232476173 = 0.980993232476182$$

```
>
> 1/k*Int(t^(-(k-1)/k)*(1-1/2*t)^(1/k),t = 0 .. 1):
  expand(%): combine(%, power);
  ``=value(%); #simplify(%);
```

$$\frac{1}{k}\int_0^1 \left(1-\frac{t}{2}\right)^{\left(\frac{1}{k}\right)} t^{\left(-1+\frac{1}{k}\right)}\,dt$$

$$= \mathrm{hypergeom}\left(\left[\frac{1}{k}, -\frac{1}{k}\right], \left[1+\frac{1}{k}\right], \frac{1}{2}\right)$$

```
> 1/k*Int(t^(-1+1/k)*(1-1/2*t)^(1/k),t = 0 .. 1):
  Change(%, t=2*r, r):
  expand(%): combine(%, power);
  ``=value(%): combine(%, power);
```

$$\frac{2^{\left(\frac{1}{k}\right)}}{k}\int_0^{1/2} r^{\left(-1+\frac{1}{k}\right)}(1-r)^{\left(\frac{1}{k}\right)}\,dr$$

$$= \mathrm{hypergeom}\left(\left[\frac{1}{k}, -\frac{1}{k}\right], \left[1+\frac{1}{k}\right], \frac{1}{2}\right)$$

```
> #FunctionAdvisor(Beta);
  FunctionAdvisor(Beta, "integral_form"):
  op(1, %):
  Change(%, _k1=r, r);
  value(%);
Warning: when the "integral" form of a function is requested, only the function
name (e.g., Beta) - is expected. Extra arguments are being ignored.
```

$$B(x, y) = \int_0^1 r^{(x-1)}(1-r)^{(y-1)}\,dr$$

$$B(x, y) = \frac{\Gamma(y)\,\Gamma(x)}{\Gamma(x+y)}$$

```
>
> B:=(s,x,y) -> Int(r^(x-1)*(1-r)^(y-1),r = 0 .. s);
  ``;
  'B(1,x,y)=Beta(x,y)'; value(%): convert(%, GAMMA): is(%);
```

$$B := (s, x, y) \to \int_0^s r^{(x-1)}(1-r)^{(y-1)}\,dr$$

$$B(1, x, y) = B(x, y)$$

true

```
>
> J:= (s,x,y) -> B(s,x,y)/Beta(x,y);
```

$$J := (s, x, y) \to \frac{B(s, x, y)}{B(x, y)}$$

```
> 'B(1/2,1/k,1+1/k)';
  ``=value(%): simplify(%);
```

$$B\left(\frac{1}{2}, \frac{1}{k}, 1+\frac{1}{k}\right)$$

$$= 2^{\left(-\frac{1}{k}\right)} k\,\mathrm{hypergeom}\left(\left[\frac{1}{k}, -\frac{1}{k}\right], \left[\frac{k+1}{k}\right], \frac{1}{2}\right)$$

```
>
> # http://dlmf.nist.gov/8.17, Identität 8.17.21

  'J(1/2,1/k,1/k) = J(1/2,1/k,1+1/k) - (1/2)^(1/k)*(1/2)^(1/k)/ (1/k) /
  Beta(1/k,1/k)';

  value(%): convert(%, GAMMA):
  isolate(%, hypergeom):
  expand(%):
  simplify(%) assuming 0<k:
  E2:=combine(%, power);
  #eval(%, k=5): evalf(%);
```

$$J\left(\frac{1}{2}, \frac{1}{k}, \frac{1}{k}\right) = J\left(\frac{1}{2}, \frac{1}{k}, 1+\frac{1}{k}\right) - \frac{\left(\left(\frac{1}{2}\right)^{\left(\frac{1}{k}\right)}\right)^2 k}{B\left(\frac{1}{k}, \frac{1}{k}\right)}$$

$$E2 := \mathrm{hypergeom}\left(\left[\frac{1}{k}, -\frac{1}{k}\right], \left[\frac{k+1}{k}\right], \frac{1}{2}\right) = \frac{2^{\left(-\frac{k+1}{k}\right)}\left(k\,\Gamma\left(\frac{2+k}{2k}\right) + \Gamma\left(\frac{1}{k}\right)\sqrt{\pi}\right)}{k\,\Gamma\left(\frac{2+k}{2k}\right)}$$

```
>
> E1;
  E2;
```

```
%%-%;
simplify(%);
```

$$\mathrm{hypergeom}\left(\left[\frac{1}{k}, -\frac{1}{k}\right], \left[\frac{k+1}{k}\right], \frac{1}{2}\right) = 2^{\left(-\frac{k+1}{k}\right)} + \frac{2^{\left(-\frac{k-1}{k}\right)} \Gamma\left(\frac{k+1}{k}\right)^2}{\Gamma\left(\frac{2+k}{k}\right)}$$

$$\mathrm{hypergeom}\left(\left[\frac{1}{k}, -\frac{1}{k}\right], \left[\frac{k+1}{k}\right], \frac{1}{2}\right) = \frac{2^{\left(-\frac{k+1}{k}\right)} \left(k\,\Gamma\left(\frac{2+k}{2\,k}\right) + \Gamma\left(\frac{1}{k}\right)\sqrt{\pi}\right)}{k\,\Gamma\left(\frac{2+k}{2\,k}\right)}$$

$$0 = 2^{\left(-\frac{k+1}{k}\right)} + \frac{2^{\left(-\frac{k-1}{k}\right)} \Gamma\left(\frac{k+1}{k}\right)^2}{\Gamma\left(\frac{2+k}{k}\right)} - \frac{2^{\left(-\frac{k+1}{k}\right)} \left(k\,\Gamma\left(\frac{2+k}{2\,k}\right) + \Gamma\left(\frac{1}{k}\right)\sqrt{\pi}\right)}{k\,\Gamma\left(\frac{2+k}{2\,k}\right)}$$

$$0 = 0$$

>